Symbolic Computation of Nonlocal Symmetries and Nonlocal Conservation Laws of Partial Differential Equations Using the GeM Package for Maple

Alexei F. Cheviakov

Abstract The use of the symbolic software package `GeM` for `Maple` is illustrated with examples of computation of nonlocal symmetries and nonlocal conservation laws of nonlinear partial differential equations. In the considered examples, the nonlocal symmetries and conservation laws arise as local symmetries and conservation laws of potential systems. Full `Maple` code with detailed comments is presented. Examples of automated symmetry and conservation law classification are included.

1 Introduction

The majority of contemporary mathematical models involving partial and ordinary differential equations (PDE, ODE) are essentially nonlinear. The analysis of such models often proceeds using approximate, numerical, and/or problem-specific methods. In particular, the efficiency and precision of numerical solutions is commonly restricted by nonlinear effects, which limit mesh sizes and boost computation times, as well as by extra large data structures arising in discretizations of multi-dimensional problems.

Methods based on the framework of symmetry and conservation law analysis can be systematically applied to wide classes of PDE and ODE models. This research area, pioneered by Sophus Lie and Emmy Noether, has been recently developed in various directions, having become a set of interrelated methods that can provide essential analytical information about the underlying equations. For further details, an interested reader is referred to [6, 9, 10, 15, 26, 33].

For ODEs, seeking conservation laws is equivalent to seeking integrating factors; conserved quantities (first integrals) lead to the reduction of order. Conser-

A. F. Cheviakov (✉)
Department of Mathematics and Statistics, University of Saskatchewan,
Saskatoon, SK S7N 5E6, Canada
e-mail: cheviakov@math.usask.ca

J.-F. Ganghoffer and I. Mladenov (eds.), *Similarity and Symmetry Methods*, Lecture Notes in Applied and Computational Mechanics 73, DOI: 10.1007/978-3-319-08296-7_3,

vation laws (divergence forms) of governing PDEs yield local densities conserved by the process, as well as global conserved quantities under appropriate boundary conditions. Local conservation laws of PDEs are also used in existence, uniqueness and stability analysis (e.g., [4, 27, 31]). An important application area of local conservation laws of PDE systems is numerical simulation. Many modern numerical methods, such as finite volume, finite element, discontinuous Galerkin methods, etc., (see, e.g., [25, 28, 29]) rely on the divergence forms of the given equations.

Local symmetries of ODEs lead to the reduction of order, and can be used for the construction of particular symmetry-invariant solutions (see, e.g., [7, 12]). Depending on the structure of the symmetry Lie algebra, the knowledge of an r-parameter Lie group of point symmetries of an ODE can lead to the reduction of order by up to r.

One of the most important applications of local symmetries to nonlinear PDEs is the construction of exact solutions. This includes obtaining new solutions from known ones through the symmetry mapping, and the construction of symmetry-invariant solutions, in particular, physically important traveling wave and self-similar solutions. Additional exact solutions can be obtained using nonlocal symmetries, when they are known. Multiple examples can be found in [10] and references therein.

If a PDE system has an infinite set of local symmetries and/or local conservation laws involving arbitrary functions, it can sometimes be mapped into a linear PDE system by an invertible transformation [3, 13]. Similarly, infinite families of nonlocal symmetries and/or nonlocal conservation laws admitted by a PDE system may be used to construct respective non-invertible mappings [10, 14]. An infinite countable set of local conservation laws may be associated with integrability.

An important application of local conservation laws is the construction of potential systems, nonlocally related to a given one, through the introduction of nonlocal potential variables. Other types of nonlocally related systems, in particular, nonlocally related subsystems, can also arise. The resulting framework of nonlocally related PDE systems [8, 10, 11] has been successfully used in multiple applications, yielding nonlocal symmetries and conservation laws, nonlocal linearizations, and new classes of exact solutions of various PDE systems (see, e.g., [10] and references therein).

The systematic computation of symmetries and conservation laws of PDE systems, especially symmetry and conservation law *classifications* and case splitting for systems involving arbitrary functions or constant parameters, may present a significant computational challenge. Indeed, systems of symmetry and conservation law determining equations can involve thousands of linear PDEs. Symbolic computation software is routinely used to carry out such computations. A number of symbolic software packages have been written for local symmetry and conservation law computations in various computer algebra systems. In the current paper, the use of `GeM` package for `Maple`, developed by the author, is discussed ([22–24]). The current version 32.02 of the `GeM` package has been tested to work with `Maple` versions 14–18.

The present contribution is devoted to practical aspects of computation of nonlocal symmetries and nonlocal conservation laws of nonlinear PDEs. After the general introduction and definitions of Sect. 2, in Sect. 3, we present basic detailed examples of the use of GeM package to compute nonlocal symmetries and nonlocal conservation laws of nonlinear PDEs through the local symmetry and conservation law computations applied to potential systems. In particular, a nonlocal symmetry for a specific nonlinear wave equation is derived; nonlocal symmetries of a class of nonlinear telegraph equations are classified; nonlocal conservation laws are sought for a class of diffusion-convection equations.

The paper is concluded with Sect. 4 containing a discussion and further remarks.

2 Nonlocal Symmetries and Nonlocal Conservation Laws

Consider a system $\mathbf{R}\{x; u\}$ of N differential equations of order k, with n independent variables $x = (x^1, \dots, x^n)$ and m dependent variables $u(x) = (u^1(x), \dots, u^m(x))$, given by

$$R^\sigma[u] \equiv R^\sigma(x, u, \partial u, \dots, \partial^k u) = 0, \qquad \sigma = 1, \dots, N. \tag{1}$$

Here and below, the notation $f[u]$ denotes a differential function depending on x, u and the derivatives of u up to some finite order,

$$\partial u \equiv \partial^1 u = \left(u^1_1(x), \dots, u^1_n(x), \dots, u^m_1(x), \dots, u^m_n(x)\right)$$

denotes the set of all first-order partial derivatives, and

$$\begin{aligned}\partial^p u &= \left\{u^\mu_{i_1\dots i_p}\ ;\ \ \mu = 1, \dots, m;\ \ i_1, \dots, i_p = 1, \dots, n\right\}\\ &= \left\{\frac{\partial^p u^\mu(x)}{\partial x^{i_1} \dots \partial x^{i_p}}\ ;\ \mu = 1, \dots, m; i_1, \dots, i_p = 1, \dots, n\right\}\end{aligned}$$

denote higher-order derivatives. Summation in any pair of repeated indices is assumed below. Subscripts are used to denote partial derivatives: $u_x \equiv \partial u/\partial x$, etc.

2.1 Lie Point Symmetries

Consider a one-parameter Lie group of point transformations

$$\begin{aligned}(x^*)^i &= f^i(x, u; \varepsilon) = x^i + \varepsilon\xi^i(x, u) + O(\varepsilon^2), \qquad i = 1, \dots, n\\ (u^*)^\mu &= g^\mu(x, u; \varepsilon) = u^\mu + \varepsilon\eta^\mu(x, u) + O(\varepsilon^2), \qquad \mu = 1, \dots, m\end{aligned} \tag{2}$$

with the corresponding infinitesimal generator

$$\mathrm{X} = \xi^i(x, u)\frac{\partial}{\partial x^i} + \eta^\mu(x, u)\frac{\partial}{\partial u^\mu}. \tag{3}$$

Definition 1 The one-parameter Lie group of point transformations (2) *leaves the* DE *system* (1) *invariant* if it maps any family of solution surfaces $u = u(x)$ of the DE system (1) into another family of solution surfaces $u^* = u^*(x^*)$ of DE system (1). In this case, the transformation (2) are referred to as a *point symmetry* of the DE system (1).

The Lie's algorithm for finding the point symmetries of a DE system (1) written in a solved form in terms of a set of leading derivatives is based on the following theorem (for details, see, e.g., [10, 15, 33]).

Theorem 1 *Let* (3) *be the infinitesimal generator of a one-parameter Lie group of point transformations* (2)*, and* $\mathrm{X}^{(k)}$ *its* kth *extension. Then the transformation* (2) *is a point symmetry of the* DE *system* (1) *if and only if for each* $\alpha = 1, \dots, N$

$$\mathrm{X}^{(k)} R^\alpha(x, u, \partial u, \dots, \partial^k u) = 0 \tag{4}$$

when

$$R^\sigma(x, u, \partial u, \dots, \partial^k u) = 0, \qquad \sigma = 1, \dots, N. \tag{5}$$

In (4), it is assumed that (5) and the differential consequences of (5) are taken into account.

Other types of local symmetries, including contact and higher-order symmetries, can be computed in a similar manner, when they exist. For such extensions, the symmetry components may depend on derivatives (e.g., [10, 15, 33]).

Remark 1 It is important to mention that some PDE systems have an infinite number of local symmetries, with symmetry components involving arbitrary functions of one or more variables. In particular, linear PDEs always admit "trivial" symmetries

$$\mathrm{X}_\infty = \sigma^\mu \frac{\partial}{\partial u^\mu}$$

where a set of functions $\sigma(x) = (\sigma^1(x), \dots, \sigma^m(x))$ is an arbitrary solution of the homogeneous version of the given linear equations.

Conversely, if a given PDE system has a sufficiently large infinite set of local symmetries, it can be mapped into a linear system with a point transformation. For details on necessary and sufficient conditions for the existence of such mappings, see [5, 10, 13].

In practical symmetry computations for linear PDEs, the presence of the "trivial" infinite-dimensional symmetry groups poses certain difficulties; for details and techniques of such computations, see [24].

2.2 Local Conservation Laws

Definition 2 *A local divergence-type conservation law* of a PDE system (1) is a divergence expression of the form

$$\mathrm{D}_i \Phi^i(x, u, \partial u, \ldots, \partial^r u) = 0 \tag{6}$$

in terms of total derivative operators

$$\mathrm{D}_i = \frac{\partial}{\partial x^i} + u_i^\mu \frac{\partial}{\partial u^\mu} + u_{ii_1}^\mu \frac{\partial}{\partial u_{i_1}^\mu} + u_{ii_1i_2}^\mu \frac{\partial}{\partial u_{i_1i_2}^\mu} + \cdots \tag{7}$$

holding on solutions of (1).

In the 1+1-dimensional situation, with $x = (x, t)$, the conservation law (6) has the form

$$\mathrm{D}_t \Theta + \mathrm{D}_x \Psi = 0 \tag{8}$$

where the density Θ and the spatial flux Ψ can depend on independent and dependent variables of the given equations, as well as their derivatives.

Remark 2 In practice, one is interested in finding sets of non-trivial, non-equivalent, linearly independent conservation laws. A trivial conservation law of a normal PDE system is a divergence expression that vanishes identically, or if its density and fluxes vanish on solutions of the given PDE system. For further details, see, e.g., [10, 33].

Local conservation laws (6), (8) are systematically sought by applying the *direct conservation law construction method* [2]. The method consists in finding sets of multipliers $\{\Lambda_\sigma[U]\}_{\sigma=1}^N = \{\Lambda_\sigma(x, U, \partial U, \ldots, \partial^\ell U)\}_{\sigma=1}^N$, depending on some prescribed independent and dependent variables and possibly their derivatives to some finite order ℓ, which, taken in linear combinations with the given PDEs, yield a divergence expression

$$\Lambda_\sigma[U] R^\sigma[U] \equiv \mathrm{D}_i \Phi^i[U] \tag{9}$$

holding for arbitrary functions U. Then on solutions $U = u(x)$ of the PDE system (1), one has a local conservation law

$$\Lambda_\sigma[u] R^\sigma[u] = \mathrm{D}_i \Phi^i[u] = 0. \tag{10}$$

Determining equations for the multipliers are obtained from the fact that an expression $F(U)$ is annihilated by Euler operators

$$\begin{aligned} &\mathrm{E}_{U^j} = \frac{\partial}{\partial U^j} - \mathrm{D}_i \frac{\partial}{\partial U_i^j} + \cdots + (-1)^s \mathrm{D}_{i_1} \ldots \mathrm{D}_{i_s} \frac{\partial}{\partial U_{i_1 \ldots i_s}^j} + \cdots \\ &i, i_q = 1, \ldots, n, \qquad j = 1, \ldots, m \end{aligned} \tag{11}$$

if and only if $F(U)$ is a divergence expression (e.g., [10, 33]). Hence the local conservation law multiplier determining equations are given by

$$\mathrm{E}_{U^j}\big(\Lambda_\sigma[U]R^\sigma[U]\big) = 0, \qquad j = 1, \ldots, m. \tag{12}$$

After the linear equations (12) are solved for the multipliers $\{\Lambda_\sigma[U]\}_{\sigma=1}^N$, the conservation law fluxes and/or density is calculated using (9) (see, e.g., [23]).

Remark 3 Some PDE systems admit an infinite number of independent local conservation laws. In such cases, multipliers may involve arbitrary functions of one or several variables. This happens for both nonlinear and linear PDEs. In particular, linear PDEs always admit an infinite number of conservation laws; the corresponding conservation law multipliers are solutions of a linear system of PDEs adjoint to the given linear system (see, e.g., [5, 10]).

When a given PDE system admits a sufficiently large infinite set of local conservation laws, it can be mapped into a linear system with a point transformation, see [3, 10].

Remark 4 Local variational symmetries and local conservation laws of self-adjoint (variational) PDEs are related through the Noether's theorem. For non-variational PDE systems, this relation generally does not hold. The direct conservation law construction method described above is applicable to both variational and non-variational PDE systems [9, 10, 17, 33].

2.3 Nonlocally Related PDE Systems

Consider a PDE system $\mathbf{R}\{x, t; u\}$ with two independent variables (x, t) and m dependent variables $u = (u^1, \ldots, u^m)$ given by

$$R^\sigma[u] = R^\sigma(x, t, u, \partial u, \partial^2 u, \ldots, \partial^l u) = 0, \quad \sigma = 1, \ldots, s. \tag{13}$$

Suppose that the PDE system (13) has one or more nontrivial conservation laws (8). For each such conservation law, one can introduce a potential variable v satisfying

$$v_x = \Theta[u], \qquad v_t = -\Psi[u]. \tag{14}$$

The potential variable v is a *nonlocal variable* of the PDE system (13), i.e., v cannot be expressed as a local function of the variables in the PDE system (13) and their derivatives [10].

A *potential system* is obtained by appending one or more sets of potential equations (14) to the given PDE system (13). We denote a potential system involving q potential variables by $\mathbf{S}\{x, t; u, v\}$, $v = (v^1, ..., v^q)$.

Remark 5 In the case of PDE systems involving $n \geq 3$ independent variables, the application of divergence-type local conservation laws to the construction of potential systems is less straightforward. In particular, the corresponding potential system is underdetermined. Overdetermined potential systems cannot yield nonlocal symmetries [1], but can yield nonlocal conservation laws [1, 21]. The gauge freedom may be eliminated using a gauge constraint, however, finding an"optimal" gauge for a specific PDE system—conservation law pair remains an open problem.

2.4 Nonlocal Symmetries and Nonlocal Conservation Laws

Consider a given PDE system $\mathbf{R}\{x, t; u\}$ (13) and its potential system $\mathbf{S}\{x, t; u, v\}$ involving a single potential variable: (13), (14). Point symmetries of the potential system $\mathbf{S}\{x, t; u, v\}$ are given by infinitesimal generators

$$\mathrm{X} = \xi^x(x, t, u, v)\frac{\partial}{\partial x} + \xi^t(x, t, u, v)\frac{\partial}{\partial t} + \sum_{i=1}^{m} \eta^{u^i}(x, t, u, v)\frac{\partial}{\partial u^i} + \eta^v(x, t, u, v)\frac{\partial}{\partial v}. \tag{15}$$

Definition 3 A generator (15) corresponds to a *nonlocal symmetry* of the given PDE system $\mathbf{R}\{x, t; u\}$ (13) if it does not yield a local symmetry of (13) when projected on the space of its variables.

The criterion for the symmetry (15) to be a nonlocal symmetry of the system $\mathbf{R}\{x, t; u\}$ (13) is provided by the following theorem (e.g, [10, 15, 16]).

Theorem 2 *The point symmetry* (15) *of the potential system* $\mathbf{S}\{x, t; u, v\}$ *yields a nonlocal symmetry* (potential symmetry) *of the given PDE system* (13) *if and only if one or more of the infinitesimals* $(\xi^x(x, t, u, v), \xi^t(x, t, u, v), \eta^{u^1}(x, t, u, v), \ldots, \eta^{u^m}(x, t, u, v))$ *depend explicitly on the potential variable* v*, i.e.,*

$$\left(\frac{\partial \xi^x}{\partial v}\right)^2 + \left(\frac{\partial \xi^t}{\partial v}\right)^2 + \sum_{i=1}^{m}\left(\frac{\partial \eta^{u^i}}{\partial v}\right)^2 > 0.$$

Remark 6 Nonlocal symmetries can also arise from nonlocally related subsystems obtained by differential exclusions of dependent variables, and from other PDE systems in the trees of nonlocally related PDE systems. For details, examples, and applications, see [10] and references therein.

Now consider a local conservation law

$$\mathrm{D}_t\Theta[u, v] + \mathrm{D}_x\Psi[u, v] = 0 \tag{16}$$

of the potential system $\mathbf{S}\{x, t; u, v\}$.

Definition 4 A nontrivial local conservation law (16) of the potential system $\mathbf{S}\{x, t; u, v\}$ is called a *nonlocal conservation law of the given PDE system* $\mathbf{R}\{x, t; u\}$ if it is not equivalent to any linear combination of local conservation laws of $\mathbf{R}\{x, t; u\}$ and trivial conservation laws, i.e., the flux and/or density in (16) have an essential dependence on the components of the potential variable v.

The following fundamental theorem [11, 32] holds.

Theorem 3 *Each conservation law of any potential system* $\mathbf{S}\{x, t; u, v\}$*, arising from multipliers that do not essentially depend on the potential variable* v*, is equivalent to a local conservation law of the given system* $\mathbf{R}\{x, t; u\}$ (13).

It follows that in order to construct nonlocal conservation laws of the original system using the direct method, one must consider multipliers that essentially involve potential variable(s). A similar theorem holds for equations with three or more independent variables [10].

The procedure of construction of *an extended tree of nonlocally related* PDE *systems*, starting from a given PDE system (13), is presented in [8, 10, 11]. It is based on the systematic construction, or a given system (13) on local conservation laws, potential systems, further local and nonlocal conservation laws, further potential systems, subsystems, and so on. Similar constructs in multi-dimensions are discussed in [10, 20, 21].

Remark 7 It is important to note that in practice, nonlocal symmetries and nonlocal conservation laws usually arise in classifications when given systems involve arbitrary (constitutive) functions or constant parameters, for special cases of those constitutive functions/parameters. Many examples of such classifications can be found in [10].

Remark 8 Similarly to local symmetries and conservation laws, infinite sets of nonlocal symmetries and conservation laws can lead to a linearization by a nonlocal transformation (e.g., [10, 14]). For example, this is the case for all 1+1-dimensional nonlinear wave equations $u_{tt} = (c^2(u)u_x)_x$, whose basic potential system is linearizable by a hodograph transformation, and for the specific instances of the nonlinear telegraph equation considered in Sect. 3.2 below.

3 Symbolic Computations of Nonlocal Symmetries and Nonlocal Conservation Laws

3.1 Example 1: Local and Nonlocal Symmetry Analysis of a Nonlinear Wave Equation

Consider a nonlinear wave equation on $u = u(x, t)$, denoted by $\mathbf{R}\{x, t; u\}$

$$u_{tt} = (c^2(u)u_x)_x. \tag{17}$$

For simplicity, we restrict to a specific case

$$c^2(u) = \frac{1}{u^2+1}. \tag{18}$$

for which nonlocal symmetries are known to arise (see, e.g., [10], Sec. 4.2.2).
(A) Point symmetries. We start from point symmetry analysis of the wave equation (17) with (18). The command sequence for the `GeM` package, version 32.02, and the output proceeds as follows.

All variables are cleared `GeM` package is initialized with the command

```
restart; read("d:/gem32_02.mpl"):
```

Independent and dependent variables are put together, for convenience, by commands

```
ind:=x,t; all_dep:=U(ind);
```

In the absence of arbitrary constants and/or functions in the given equation, the variables are declared as follows:

```
gem_decl_vars(indeps=[ind], deps=[all_dep]);
```

The wave speed is defined by

```
c(U(ind)):=1/(U(ind)^2+1);
```

and the given PDE is further defined, in the solved form, as follows:

```
gem_decl_eqs([diff(U(ind),t,t)=diff(c(U(ind))^2*diff(U(ind),x),x)],
             solve_for=[diff(U(ind),t,t)]);
```

The split system of linear symmetry determining equations, where the symmetry components depend on all independent and dependent variables, is generated by the function

```
det_eqs:=gem_symm_det_eqs([ind, all_dep]):
```

yielding 16 determining equations. A variable containing the unknown symmetry components will be needed for the further computations. It is initialized using the function

```
sym_components:=gem_symm_components();
```

The output value is

sym_components := [xi_t(x, t, U), xi_x(x, t, U), eta_U(x, t, U)]

where the three quantities correspond to the symmetry components for t, x, u respectively.

The simplification and differential elimination of redundant determining equations is obtained by calling the Maple rifsimp routine, as follows

```
symm_det_eqs:=DEtools[rifsimp](det_eqs, sym_components, mindim=1);
```

The mindim=1 option will force rifsimp to output the dimension of the solution space, i.e., the number of independent point symmetries of the PDE (17). In this case, the returned dimension is three. The system of 16 determining equations is reduced to seven equations and is stored in symm_det_eqs[Solved]. The final solution is performed using the standard Maple pdsolve routine,

```
symm_sol:=pdsolve(symm_det_eqs[Solved],sym_components);
```

returning

*symm_sol:= eta_U(x, t, U) = 0, xi_t(x, t, U) = _C1*t+_C3, xi_x(x, t, U)= _C1*x+_C2*

This final solution involves three arbitrary constants *_C1, _C2, _C3*, which agrees with the dimension of the solution space returned by rifsimp.

Finally, the three independent symmetry generators are output using the command

```
gem_output_symm(symm_sol);
```

which yields the canonical forms of the point symmetries of the PDE (17) with the wave speed (18)

$$\mathrm{X}_1 = \frac{\partial}{\partial t}, \qquad \mathrm{X}_2 = \frac{\partial}{\partial x}, \qquad \mathrm{X}_3 = t\frac{\partial}{\partial t} + x\frac{\partial}{\partial x}.$$

The equation is thus invariant under $t-$ and $x-$ translations and a scaling.

(B) A nonlocal symmetry computation. For an arbitrary $c(u)$, the PDE (17) has four zeroth-order conservation laws with multipliers $\Lambda = 1, t, xt, x$. In this example, we use the second one. The conservation law is given by

$$\mathrm{D}_t(tu_t - u) - \mathrm{D}_x(tc^2(u)u_x) = 0$$

and the resulting potential system $\mathbf{S}\{x, t; u, w\}$ is

$$w_x = tu_t - u, \quad w_t = tc^2(u)u_x. \tag{19}$$

We study local symmetries of (19) to seek nonlocal symmetries of the PDE (17) with the wave speed (18). The program proceeds in a fashion similar to the above example.

```
restart; read("d:/gem32_02.mpl"):
ind:=x,t; all_dep:=U(ind),W(ind);
gem_decl_vars(indeps=[ind], deps=[all_dep]);
c(U(ind)):=1/(U(ind)^2+1);
```

```
gem_decl_eqs([diff(W(ind),x)=t*diff(U(ind),t)-U(ind),
             diff(W(ind),t)=t*c(U(ind))^2*diff(U(ind),x)],
             solve_for=[diff(W(ind),x),diff(W(ind),t)]);

det_eqs:=gem_symm_det_eqs([ind, all_dep]):
sym_components:=gem_symm_components();
symm_det_eqs:=DEtools[rifsimp](det_eqs, sym_components, mindim=1);
symm_sol:=pdsolve(symm_det_eqs[Solved],sym_components);
gem_output_symm(symm_sol);
```

Here the dimension of the solution space is four and the output contains four symmetry generators

$$\mathrm{Y}_1 = \frac{\partial}{\partial w}, \qquad \mathrm{Y}_2 = \frac{\partial}{\partial x}, \qquad \mathrm{Y}_3 = t\frac{\partial}{\partial t} + x\frac{\partial}{\partial x} + w\frac{\partial}{\partial w}$$
$$\mathrm{Y}_4 = tu\frac{\partial}{\partial t} + w\frac{\partial}{\partial x} + (u^2+1)\frac{\partial}{\partial w} - x\frac{\partial}{\partial w}.$$

The symmetry Y_4 yields a nonlocal symmetry of the given nonlinear wave equation (17), since the component $\xi^x = w$ involves the potential variable (cf. Definition 3).

3.2 *Example 2: A Potential Symmetry Classification for the Nonlinear Telegraph Equation*

Let $\mathbf{R}\{x, t; u\}$ denote the nonlinear telegraph equation with the unknown function $u = u(x, t)$, given by

$$u_{tt} = (F(u)u_x)_x + (G(u))_x. \tag{20}$$

The complete point symmetry classification of the PDE (20) with respect to the constitutive functions $F(u)$ and $G(u)$ can be found in Ref. [30].

The PDE (20) is a conservation law as it stands, hence one can introduce a potential $v(x, t)$ to obtain a potential system

$$u_t = v_x, \qquad v_t = F(u)u_x + G(u). \tag{21}$$

The point symmetry classification of the PDE system (21) has been performed in [18]. In particular, it has been shown that in the cases $F(u) = u^{-2}$, $G(u) = u^{-1}$, and $F(u)$ arbitrary, $G(u) = \text{const}$, the potential system (21) has an infinite number of point symmetries (nonlocal symmetries of the PDE (20)), and moreover, is linearizable by a point transformation. (Thus the corresponding NLT equations (20) are linearizable by a nonlocal transformation).

The first equation of the system (21) is also a conservation law. Introducing a second potential w accordingly, one has a potential system $\mathbf{S}\{x, t; u, v, w\}$ for three dependent variables $u(x, t)$, $v(x, t)$, $w(x, t)$, given by three PDEs

$$w_t = v, \qquad w_x = u, \qquad v_t = F(u)u_x + G(u). \tag{22}$$

In our computations, for the simplicity of presentation, we will be avoiding detailed calculations for the linearization cases, since they yield infinite sets of symmetries.

For the current example, we are interested in finding point symmetries of the potential system $\mathbf{S}\{x, t; u, v, w\}$ (22) that correspond to nonlocal symmetries of the original PDE (20). For brevity, we will restrict ourselves to the case of power nonlinearities,

$$F(u) = u^{\alpha}, \quad G(u) = u^{\beta}. \tag{23}$$

The classification will thus be performed with respect to two constitutive parameters $\alpha \neq 0$, $\beta \neq 0$. (For the complete classification, see [10], Sect. 4.2.)

Point symmetry generators of the potential system $\mathbf{S}\{x, t; u, v, w\}$ (22) are of the form

$$\begin{aligned} \mathrm{Z} = \xi(x, t, u, v, w)\frac{\partial}{\partial x} + \tau(x, t, u, v, w)\frac{\partial}{\partial t} + \eta^u(x, t, u, v, w)\frac{\partial}{\partial u} \\ + \eta^v(x, t, u, v, w)\frac{\partial}{\partial v} + \eta^w(x, t, u, v, w)\frac{\partial}{\partial w}. \end{aligned} \tag{24}$$

In order to find symmetries (24) that correspond to nonlocal symmetries of the given PDE (20), one requires that at least one of the following six conditions is satisfied

$$\begin{aligned} \frac{\partial \xi^t}{\partial v} \neq 0, \qquad \frac{\partial \xi^t}{\partial w} \neq 0, \qquad \frac{\partial \xi^x}{\partial v} \neq 0 \\ \frac{\partial \xi^x}{\partial w} \neq 0, \qquad \frac{\partial \eta^u}{\partial v} \neq 0, \qquad \frac{\partial \eta^u}{\partial w} \neq 0. \end{aligned} \tag{25}$$

The `Maple` code for the symmetry classification proceeds as follows.

```
restart; read("d:/gem32_02.mpl"):

ind:=x,t; all_dep:=U(ind),V(ind),W(ind);

gem_decl_vars(indeps=[ind], deps=[all_dep],
              freeconst=[],freefunc=[F(U(ind)),G(U(ind))]);
```

Here $F(u)$ and $G(u)$ are defined as arbitrary functions. It is important to do so even though we are going to consider only power nonlinearities. The reason is that in order to do the case splitting and simplification of the determining equations, the `Maple` function `rifsimp` is used; the latter can only handle polynomial nonlinearities. Hence the code will proceed as follows

- Generate symmetry determining equations treating the nonlinear functions as arbitrary (free) at the initial stage.
- In order to use the desired form of the "arbitrary functions", employ the `Maple` `dpolyform` function. This function converts a specified condition into the differential polynomial form. An analogous operation can be performed by hand. E.g., if we need to have $H(u) = Ae^{ku}$, $A, k =$ const, the linear ODE and conditions on $H(u)$ can be

$$\frac{dH(u)}{du} = kH(u), \qquad k \neq 0, \qquad H(u) \neq 0.$$

- When case splitting with `rifsimp` is performed, the system of determining equations should be appended with the conditions defining the arbitrary functions.

The same approach should be used for any nonlinearities, including logarithms, exponents, trigonometric functions, etc.

In our case, the conditions of $F(u)$ and $G(u)$ being power nonlinearities (23) can be generated as follows. (We note that in the determining equations in `GeM`, dependent variables of the given equations are treated as simple variables, not functions; hence in the determining equations, `F(U)` not `F(U(ind))` should be used.)

```
cond_FG_powers:={F(U)=U^n, G(U)=U^m};

cond_F_G:=PDEtools[dpolyform](cond_FG_powers,no_Fn);

cond_F_G_full:=convert(cond_F_G,list)[1][], F(U)<>0, G(U)<>0,
                                                    m<>0, n<>0;
```

The resulting set of conditions is given by a `Maple` set-type variable

*cond_F_G_full := {diff(F(U), U) = F(U)*n/U, diff(G(U), U) = G(U)*m/U,*

m<>0, n<>0, F(U) <> 0, G(U) <> 0 }

The equations are declared as follows.

```
gem_decl_eqs([diff(W(x,t),t)=V(x,t),
             diff(W(x,t),x)=U(x,t),
             diff(V(x,t),t)=F(U(x,t))*diff(U(x,t),x) +G(U(x,t))],
             solve_for=[diff(W(x,t),t), diff(W(x,t),x),
             diff(V(x,t),t)]);
```

Then the symmetry determining equations are generated and a variable of symmetry components is initialized

```
det_eqs:=gem_symm_det_eqs([ind, all_dep]):
```

```
sym_components:=gem_symm_components();
```

The next step is to unite the determining equations and the conditions `cond_F_G_full` on the functions $F(u)$ and $G(u)$ to be power nonlinearities. A set union is used:

```
det_eqs:=det_eqs union cond_F_G_full:
```

We now perform six rounds the symmetry classification and case splitting, using, one by one, the conditions (25) for the symmetry to be essentially nonlocal. For the classification, it is essential to use the `casesplit` option in the call to `rifsimp`.

Round 1: $\partial\xi^t/\partial v \neq 0$.

```
symm_det_eqs:=DEtools[rifsimp](det_eqs
                union diff{xi_t(x, t, U, V, W),V)<>0},
                sym_components, mindim=1, casesplit);
```

The result contains only one case, with $m = -1, n = -2$, with the solution space of $dimension = \infty$. As remarked above, we will not go into detailed computations for this linearization case.

Round 2: $\partial\xi^t/\partial w \neq 0$.

```
symm_det_eqs:=DEtools[rifsimp](det_eqs
                union {diff(xi_t(x, t, U, V, W),W)<>0},
                sym_components, mindim=1, casesplit);
```

This computation yields two cases: the linearization case $m = -1, n = -2$ and another case $m = 3, n = 2$. (A case tree with pivots may be plotted using the command `DEtools[caseplot](symm_det_eqs,pivots);`).

Let us compute all symmetries for the case $m = 3, n = 2$. The solution space dimension (number of linearly independent symmetries) is equal to six. To obtain a general symmetry generator, one uses the following commands.

```
symm_sol:=pdsolve(
          subs({m=3,n=2,F(U)=A*U^2,G(U)=B*U^3}
          symm_det_eqs[1][Solved]), sym_components);
```

The six symmetries can be output separately using

```
gem_output_symm(symm_sol);
```

The resulting set of symmetries is given by

$$Z_1 = \frac{\partial}{\partial x}, \qquad Z_2 = \frac{\partial}{\partial t}, \qquad Z_3 = \frac{\partial}{\partial w}, \qquad Z_4 = t\frac{\partial}{\partial w} + \frac{\partial}{\partial v} + w\frac{\partial}{\partial w}$$

$$Z_5 = t\frac{\partial}{\partial t} - u\frac{\partial}{\partial u} - 2v\frac{\partial}{\partial v} - w\frac{\partial}{\partial w}$$
$$Z_6 = (Au + Bw)\frac{\partial}{\partial t} + Av\frac{\partial}{\partial x} - Buv\frac{\partial}{\partial u} - Bv^2\frac{\partial}{\partial v} + Auv\frac{\partial}{\partial w}.$$

The symmetry Z_6 yields the nonlocal symmetry of the original NLT equation (20).
Round 3: $\partial\xi^x/\partial v \neq 0$.

```
symm_det_eqs:=DEtools[rifsimp](det_eqs
              union {diff(xi_x(x, t, U, V, W),V)<>0},
              sym_components, mindim=1, casesplit);
```

This computation results in the same two cases: $m = -1$, $n = -2$ and $m = 3$, $n = 2$, no new symmetries arise.
Round 4: $\partial\xi^x/\partial w \neq 0$.

```
symm_det_eqs:=DEtools[rifsimp](det_eqs
              union {diff(xi_x(x, t, U,V,W),W)<>0},
              sym_components, mindim=1, casesplit);
```

The output of the above command is

symm_det_eqs := table([status = "system is inconsistent"*])*

which means that there are no point symmetries (24) of the potential system $\mathbf{S}\{x, t; u, v, w\}$ (22) that satisfy $\partial\xi^x/\partial w \neq 0$.
Round 5: $\partial\eta^u/\partial v \neq 0$.

```
symm_det_eqs:=DEtools[rifsimp](det_eqs
              union {diff(eta_U(x, t, U, V, W),V)<>0},
              sym_components, mindim=1, casesplit);
```

The computation again yields the same two cases: $m = -1$, $n = -2$ and $m = 3$, $n = 2$.
Round 6: $\partial\eta^u/\partial w \neq 0$.

```
symm_det_eqs:=DEtools[rifsimp](det_eqs
              union {diff(eta_U(x, t, U, V, W),W)<>0},
              sym_components, mindim=1, casesplit);
```

The result of this computations is the same as for Round 4 above: no point symmetries of the potential system $\mathbf{S}\{x, t; u, v, w\}$ (22) satisfying $\partial\eta^u/\partial w \neq 0$ exist.

3.3 Example 3: A Nonlocal Conservation Law Classification for the Nonlinear Telegraph Equation

Consider a class of diffusion-convection equations $\mathbf{R}\{x, t; u\}$ of the form

$$u_t = (A(u)u_x)_x + (B(u))_x \tag{26}$$

where $A(u)$ and $B(u)$ are arbitrary constitutive functions, and $A(u) \neq 0$. The break linear case $A = 1$, $B =$ const is excluded. The complete classification of linearly independent local conservation laws for (26) yields the following results [34].

1. For arbitrary $A(u)$, $B(u)$, the only local conservation law of (26) is given by

$$\mathrm{D}_t(u) - \mathrm{D}_x\Big(A(u)u_x + B(u)\Big) = 0. \tag{27}$$

2. For arbitrary $A(u)$, and $B(u) = 0$, there are two local conservation laws of (26).
3. For arbitrary $A(u)$, and $B(u) = A(u)$, the PDE (26) has four local conservation laws.

We employ the conservation law (27) to construct the potential system $\mathbf{S}\{x, t; u, v\}$

$$v_x = u, \qquad v_t = A(u)u_x + B(u). \tag{28}$$

We wish to perform the local conservation law classification of the potential system (28) and find conservation laws that yield nonlocal conservation laws of the given PDE (26).

Here we restrict to zeroth-order multipliers $\Lambda_1(x, t, U, V)$, $\Lambda_2(x, t, U, V)$. Moreover, for simplicity of computation, we specify

$$A(u) := u^4$$

and perform the nonlocal conservation law classification with respect to the remaining arbitrary function $B(u)$. The full `Maple` program for the computation is given below.

First, the package is initialized, and variables and the free function $B(u)$ are declared.

```
restart; read("d:/gem32_02.mpl"):
ind:=x,t; all_dep:=U(ind),V(ind);
gem_decl_vars(indeps=[ind], deps=[all_dep],freefunc=[B(U(ind))]);
```

Second, the function $A(u)$ is specialized, and the PDEs (28) are declared.

```
A(U(ind)):= (U(ind)^4);
```

```
gem_decl_eqs([diff(V(ind),x)=U(ind),
             diff(V(ind),t)=A(U(ind))*diff(U(ind),x) + B(U(ind))],
             solve_for=[diff(V(ind),x),diff(V(ind),t)]);
```

The conservation law determining equations are obtained, in the split form, by calling the functon

```
det_eqs:=gem_conslaw_det_eqs([ind, all_dep]):
```

where the list in `[...]` determines the dependence of the multipliers. This yields 6 determining equations.

The multiplier variables are accessed by calling

```
CL_multipliers:=gem_conslaw_multipliers();
```

Further, we perform the case splitting, assuming that $B(u) \neq 0$, and seeking conservation laws where at least one multiplier essentially involves the potential v (cf. Theorem 3).

Round 1: $\partial\Lambda_1/\partial V \neq 0$. Here one has

```
simplified_eqs:=DEtools[rifsimp]( {det_eqs[]} union {B(U)<>0}
               union {diff(Lambda1(x, t, U, V),V)<>0},
               CL_multipliers, casesplit, mindim=1);
```

The only nontrivial case returned has dimension one, i.e., it yields a single conservation law of the required type. The condition on the function $B(u)$ within `simplified_eqs[Solved]` is

diff(B(U), U, U, U, U) = (6(diff(B(U), U, U, U))*U−12*(diff(B(U), U, U)))/U^2*

Using `dsolve`, one readily finds that $B(U)$ must have the form

$$B(U) = M_1U^6 + M_2U^5 + M_3U + M_4, \qquad M_1, \ldots, M_4 = \text{const.}$$

For the subsequent computations, it is more straightforward to initialize the function $B(U)$ to the above expression, and then again perform the simplification of determining equations

```
B(U):=M1*U^6+M2*U^5+M3*U+M4;
```

```
simplified_eqs:=DEtools[rifsimp](det_eqs, CL_multipliers, mindim=1);
```

One then solves for the multipliers

```
multipliers_sol:=pdsolve(simplified_eqs[Solved], CL_multipliers);
```

to obtain

```
multipliers_sol :=
{Lambda1(x, t, U, V) = -_C1*exp(−(5*(M1*M4-M2*M3))*t)*exp(5*M1*V)
                                         *exp(5*M2*x)*(M1*U^5+M3)
Lambda2(x, t, U, V) =_C1*exp(-(5*(M1*M4-M2*M3))*t)
                                         *exp(5*M1*V)*exp(5*M2*x)}
```

Here $\_C1$ is the only arbitrary constant, so indeed, one conservation law is obtained. In the regular form

$$\begin{aligned}\Lambda_1(x,t,U,V) &= -C_1(M_1U^5+M_3)\mathrm{e}^{5(M_2x+M_1V+(M_2M_3-M_1M_4)t)}\\ \Lambda_2(x,t,U,V) &= C_1\mathrm{e}^{5(M_2x+M_1V+(M_2M_3-M_1M_4)t)}.\end{aligned} \tag{29}$$

Finally, the conservation law density and flux are computed using the function

```
gem_get_CL_fluxes(multipliers_sol);
```

(Other flux computation methods are available; see [10, 23].)

The newly computed conservation law is given by

$$\begin{aligned}&D_t\mathrm{e}^{5(M_2x+M_1v+(M_2M_3-M_1M_4)t)}\\ &\quad -D_x\left((M_1u^5+M_3)\mathrm{e}^{5(M_2x+M_1v+(M_2M_3-M_1M_4)t)}\right)=0.\end{aligned} \tag{30}$$

This is a nonlocal conservation law of the diffusion-convection PDE (26) since it is not equivalent to any of its local conservation laws [10, 34].

Round 2: $\partial\Lambda_2/\partial V \neq 0$. For this case, one obtains exactly the same result, i.e., a single pair of potential-dependent multipliers (29). Indeed, both multipliers there have a similar exponential dependence on v.

4 Discussion

The symbolic software package `GeM` for `Maple`, in conjunction with standard `Maple` routines like `rifsimp` and `dsolve/pdsolve`, offers convenient ways to compute symmetries and conservation laws of systems of differential equations, and importantly, perform symmetry and conservation law classifications with respect to arbitrary functions and parameters.

Such computations can be applied to potential systems, as shown in Sect. 3.1 where local symmetry analysis of a potential system for a nonlinear wave equation was used to compute a nonlocal symmetry of that equation.

It is rather straightforward, by appending extra conditions, to restrict the computations to seek specifically nonlocal symmetries or new (local or nonlocal) conservation laws arising as local ones for the potential system. In Sect. 3.2, the nonlinear telegraph equation (20) and its potential system (22) involving two potential variables were considered. By using the conditions (25) in symbolic symmetry computations for the potential system, the two cases of power nonlinearities $m = -1$, $n = -2$ and $m = 3$, $n = 2$ were isolated, for which nonlocal symmetries of the original equation arise.

In symmetry and conservation law computations in `Maple` that use `rifsimp`, nonlinearities have to be restricted to differential polynomial ones. Hence if a given DE system contains nonlinearities of other types, one has to recast the nonlinear functions into "arbitrary" functions that satisfy additional linear or differential polynomial equations. An example for power nonlinearities was presented in Sect. 3.2.

In cases when a PDE system under consideration has an infinite set of symmetries and/or conservation laws, the `rifsimp` routine returns *dimension* $= \infty$ for each such case. It is important to exercise care in the analysis of determining equations in the `rifsimp` output in such situations. In particular, `Maple pdsolve` may return incomplete results.

For PDE systems involving $n \geq 3$ independent variables, computations generally proceed the same way. Many specific aspects of construction of nonlocally related systems and nonlocal symmetry and conservation law computations were discussed in [10, 20, 21].

In terms of the further development of symbolic software for nonlocal and local symmetry/conservation law analysis and classification, the future research directions will naturally include the detailed development of more examples, and the automation of case-by-case consideration in the classifications.

Acknowledgments The author is thankful to NSERC for the research support.

References

1. Anco, S.C., Bluman, G.W.: Nonlocal symmetries and nonlocal conservation laws of Maxwell's equations. J. Math. Phys. **38**, 3508–3532 (1997)
2. Anco, S.C., Bluman, G.W.: Direct construction of conservation laws from field equations. Phys. Rev. Lett. **78**, 2869–2873 (1997)
3. Anco, S.C., Bluman, G.W., Wolf, T.: Invertible mappings of nonlinear PDEs to linear PDEs through admitted conservation laws. Acta Appl. Math. **101**, 21–38 (2008)
4. Benjamin, T.B.: The stability of solitary waves. Proc. Roy. Soc. Lond. A **328**, 153–183 (1972)
5. Bluman, G.W.: Use of group methods for relating linear and nonlinear partial differential equations. In: Proceedings of Symposium on Symmetry, Similarity and Group Theoretic Methods in Mechanics, pp. 203–218, Calgary AB (1974)
6. Bluman, G.: Connections Between Symmetries and Conservation Laws, Symm. Integr. & Geom.: Meth. Appl. (SIGMA) 1 (2005) paper 011, 16 pages
7. Bluman, G.W., Anco, S.C.: Symmetry and Integration Methods for Differential Equations for Differential Equations, Applied Mathematical Sciences. Springer, New York (2005)
8. Bluman, G., Cheviakov, A.F.: Framework for potential systems and nonlocal symmetries: algorithmic approach. J. Math. Phys. **46**, 123506 (2005)
9. Bluman, G., Cheviakov, A.F., Anco, S.: Construction of conservation laws: how the direct method generalizes Noether's theorem. In: Proceedings of 4th Workshop "Group Analysis of Differential Equations & Integrability, vol. 1, pp. 1–23 (2009)

10. Bluman, G.W., Cheviakov, A.F., Anco, S.C.: Applications of Symmetry Methods to Partial Differential Equations. Applied Mathematical Sciences. Springer, New York (2010)
11. Bluman, G., Cheviakov, A.F., Ivanova, N.M.: Framework for nonlocally related PDE systems and nonlocal symmetries: extension, simplification, and examples. J. Math. Phys. **47**, 113505 (2006)
12. Bluman, G.W., Cheviakov, A.F., Senthilvelan, M.: Solution and asymptotic/blow-up behaviour of a class of nonlinear dissipative systems. J. Math. Anal. Appl. **339**, 1199–1209 (2008)
13. Bluman, G.W., Kumei, S.: Symmetry-based algorithms to relate partial differential equations: I. local symmetries. Eur. J. Appl. Math. **1**, 189–216 (1990)
14. Bluman, G.W., Kumei, S.: Symmetry-based algorithms to relate partial differential equations: II. linearization by nonlocal symmetries. Eur. J. Appl. Math. **1**, 217–223 (1990)
15. Bluman, G.W., Kumei, S.: Symmetries and Differential Equations, Applied Mathematical Sciences. Springer, New York (1989)
16. Bluman, G.W., Kumei, S., Reid, G.J.: Classes, new, of symmetries for partial differential equations. J. Math. Phys. **29**, 806–811 (Erratum. J. Math. Phys. **29**(1988), 2320 (1988))
17. Bluman, G.W., Temuerchaolu.: Comparing symmetries and conservation laws of nonlinear telegraph equations. J. Math. Phys. **46**, 073513 (2005)
18. Bluman, G.W., Temuerchaolu, Sahadevan, R.: Local and nonlocal symmetries for nonlinear telegraph equations. J. Math. Phys. **46**, 023505 (2005)
19. Cheviakov, A.F.: Comp. Phys. Comm. **176**(1), 48–61 (2007). (In the current paper, we used a new version of `"GeM"` software, which is scheduled for public release in 2009. See http://math.usask.ca/cheviakov/gem/)
20. Cheviakov, A.F., Bluman, G.W.: Multidimensional partial differential equation systems: generating new systems via conservation laws, potentials, gauges, subsystems. J. Math. Phys. **51**, 103521 (2010)
21. Cheviakov, A.F., Bluman, G.W.: Multidimensional partial differential equation systems: nonlocal symmetries, nonlocal conservation laws, exact solutions. J. Math. Phys. **51**, 103522 (2010)
22. Cheviakov, A.F.: GeM software package for computation of symmetries and conservation laws of differential equations. Comput. Phys. Commun. **176**, 48–61 (2007)
23. Cheviakov, A.F.: Computation of fluxes of conservation laws. J. Eng. Math. **66**, 153–173 (2010)
24. Cheviakov, A.F.: Symbolic computation of local symmetries of nonlinear and linear partial and ordinary differential equations. Math. Comp. Sci. **4**, 203–222 (2010)
25. Godlewski, E., Raviart, P.-A.: Numerical Approximation of Hyperbolic Systems of Conservation Laws. Springer, Berlin (1996)
26. Hydon, P.E.: SSymmetry Methods for Differential Equations: A Beginner's Guide. Cambridge University Press, Cambridge (2000)
27. Lax, P.D.: Integrals of nonlinear equations of evolution and solitary waves. Comm. Pure Appl. Math. **21**, 467–490 (1968)
28. LeVeque, R.J.: Numerical Methods for Conservation Laws. Birkhäuser, Basel (1992)
29. Kallendorf, C., Cheviakov, A.F., Oberlack, M., Wang, Y.: Conservation laws of surfactant transport equations. Phys. Fluids **24**(10), 102105 (2012)
30. Kingston, G., Sophocleous, C.: Symmetries and form-preserving transformations of one-dimensional wave equations with dissipation. J. Non-Linear Mech. **36**, 987–997 (2001)
31. Knops, R.J., Stuart, C.A.: Quasiconvexity and uniqueness of equilibrium solutions in nonlinear elasticity. Arch. Rat. Mech. Anal. **86**, 234–249 (1984)
32. Kunzinger, M., Popovych, R.O.: Potential conservation laws. J. Math. Phys. **49**, 103506 (2008)
33. Olver, P.J.: Applications of Lie Groups to Differential Equations. Graduate Texts in Mathematics. Springer, New York (1986)
34. Popovych, R.O., Ivanova, N.M.: Hierarchy of conservation laws of diffusion-convection equations. J. Math. Phys. **46**, 043502 (2005)
35. Wolf, T.: A comparison of four approaches to the calculation of conservation laws. Europ. J. Appl. Math. **13**(2), 129–152 (2002)
36. Wolf, T.: Crack, LiePDE, ApplySym and ConLaw, section 4.3.5 and computer program on CD-ROM. In: Grabmeier J., Kaltofen E., Weispfenning, V. (eds.) Computer Algebra Handbook. pp. 465–468, Springer (2002)